

CALCULATION COMPUTER PROGRAM FOR A QUASI-ADIABATIC CALORIMETER

F. RODANTE, A. ONOFRI and P. PERTICAROLI

*Dipartimento di Ingegneria Chimica, dei Materiali, delle Materie Prime e Metallurgia,
Università "La Sapienza", Via del Castro Laurenziano 7, 00161 Roma (Italy)*

(Received 4 May 1987)

ABSTRACT

A computer program (in the language BASIC) for the determination of solution and reactions heats in a quasi-adiabatic calorimeter has been presented. The menu of the program is made up of the following steps: (1) calibration curve; (2) cooling curve; (3) reaction curve; and (4) equilibrium temperature.

There are also three subroutines which are able to modify the values of time–voltage pairs for steps (1)–(3), to change the data of the reaction curve (K/C , instantaneous power and mole number) and to reorganize the order of time–voltage pairs for steps (1)–(3) as a function of increasing time vector.

The program was tested using the standard thermal reaction of solid tris(hydroxymethyl)amino methane (THAM) with 0.1 M aqueous HCl. The value obtained is in good agreement with that obtained in the literature.

INTRODUCTION

The study of a calculation method for the determination of solution and reaction heats in a constant-environment or in an adiabatic calorimeter has been the subject of a recent paper [1]. In that work the calculation of a corrective term for the temperature change connected with the heat exchange of the vessel with the surroundings has been particularly stressed.

Further contribution to this study has been brought about by a computer program (in the language BASIC) for the calculation of solution and reaction heats in a constant-environment calorimeter [2]. Particular attention has been paid to the choice of parameters such as K/C (where K is the "leakage constant" and C the heat capacity of the system) and the end-point of a generic calorimetric reaction in liquid phase.

This work aims at supplying a computer program for an isoperibol (quasi-adiabatic) calorimeter and at stressing that this calculation procedure is useful for both types of calorimeters.

EXPERIMENTAL

The Tronac (model 458) calorimeter was used. This is a quasi-adiabatic calorimeter especially suitable for measuring the heats developed by reagents in liquid phase. With regard to this, thermometric titrations have been used for the simultaneous determination of both reaction enthalpies and reaction constant values (free energies). In this way, the entropies of the reactions were also calculated. This technique is known as "entropic titration" [3-5]. The same authors have carried out a computer program for the calculation of heats of reactions and equilibrium constants (FORTRAN language) [6]. Other programs (BASIC language) for the refinement of thermodynamic parameters obtained from thermometric titrations are also available [7-10]. The calorimeter vessel was a rapid-response, glass vacuum Dewar of 100 cm³ maximum capacity. The burette volume was 2 cm³ and the measured titrant delivery rate 0.49373 cm³ min⁻¹. The thermostat was maintained at 298.16 ± 0.0002 K by employing a Tronac P.T.C.-41 precision temperature controller. Potential vs. time measurements were made using a Fluke 8810 A model digital voltmeter. To measure the temperature change a thermistor with a value of 1974 ohms at 25°C was used. Its experimental calibration (T vs. mV) indicates that it follows the linear equation $T = 3.16 \times 10^{-6} mV + 25$. The unbalance of the bridge is fed into an Hitachi 561-10002/P strip-chart recorder and into the digital voltmeter, which in turn is connected to an Olivetti M24 Computer. The BASIC language was used.

As previously shown [1], the method of heats calculation was founded on the hypothesis that the calorimeter really exhibited adiabatic behaviour throughout the reaction. By knowing the temperature change ΔT_{adb} and the heat capacity C of the adiabatic system, it is possible to calculate the heat absorbed or developed during the reaction using the equation

$$Q = C\Delta T \quad (1)$$

In the real trend with heat exchange between the vessel and the surroundings the adiabatic value temperature is given by the following reaction

$$\Delta T_{\text{adb}} = \Delta T_{\text{obs}} + \Delta T_{\text{corr}} \quad (2)$$

where ΔT_{obs} is the observed value and ΔT_{corr} is the correction term which bears in mind the fact that during the reaction the vessel temperature changes by virtue of heat exchange with the surroundings. The ΔT_{corr} can be written as [1]

$$\Delta T_{\text{corr}} = \int_{t_0}^t (T_t - T_\infty) dt \quad (3)$$

(where t_0 is the time at which the reaction starts and t any time after the end of the reaction, T_t is the temperature trend inside the vessel (as a

function of the time), and T_∞ the constant temperature approached by the calorimeter vessel after a very long time.

PROGRAM DESCRIPTION

The menu of the program is made up of the following steps: (1) calibration curve; (2) cooling curve; (3) reaction curve; and (4) equilibrium temperature. There are also three subroutines. The first allows modification of the values of the time (t)–voltage (V) pairs for steps (1)–(3). The second allows changes to K/C , the instantaneous power and the mole number of the reaction curve. The third reorganizes the positions of the time–voltage pairs for steps (1)–(3) as a function of time.

Calibration curve

This part allows calculation of the temperature rise, as a function of the time, dT/dt , during the electrical calibration. Six time–voltage pairs every 30 s are sufficient for the calibration. It is necessary to choose these pairs after the transition periods of the heat conduction are completed. The program allows the taking of a hundred time–voltage pairs.

Cooling curve

In this part of the program, the K/C coefficient is calculated. This may be realized bearing in mind that, had the calorimeter been of the adiabatic type, the ΔT_{adb} , after the end of the calibration or the chemical reaction would have been equal to zero. Thus, imposing this condition in eqn. (2) one can obtain

$$\Delta T_{\text{obs}} = -\Delta T_{\text{corr}} \quad (5)$$

and again

$$T_0 - T_\infty = -K/C \int_{t_\infty}^{t_0} (T - T_\infty) dt \quad (6)$$

with ΔT_{corr} calculated between a generic time t_0 and an infinite time t_∞ .

Equation (6) provides a differential equation, which solved with the initial condition $T_{t_0} = T_0$ gives the solution

$$T_0 - T_\infty = (T_0 - T_\infty) e^{-K/C(t-t_0)} \quad (7)$$

This last shows that the temperature approaches T_∞ as time tends to infinity, with exponential behaviour dependent on K/C .

The K/C term can be obtained as

$$K/C = -\ln(T - T_\infty / T_0 - T_\infty) / (t - t_0) \quad (8)$$

This relation can be considered like a linear equation, so that only two points need to be used. One hundred pairs of time–voltage values could be acquired, but five pairs (following each other) every six minutes, taken after the electrical calibration or the reaction effects were over, are sufficient to determine K/C .

Reaction curve

This part carries out the following steps: (a) calculation of the corrective temperature term

$$\Delta T = K/C \int_{t_0}^t (T_t - T_\infty) dt \quad (9)$$

(b) calculation of the heat capacity by means of the expression

$$C = P_{\text{cal}}/dT/dt + K/C \times (T - T_\infty) \quad (10)$$

where P_{cal} is the instantaneous power which is measured by reading the calibration current (obtained by dividing the V_S potential of the standard resistance by 100) and the potential drop V_H of calibration heater resistance. The method commonly used for determining the heat capacity is based on the supply of a known amount of energy during time t across the heater resistance.

However, this method has two drawbacks: (1) The energy supplied during t is not constant because the resistance value is a function of time; and (2) one must wait for the transition periods of the heat conduction to be completed before measuring the temperature change ΔT . This requires larger measuring times, which in turn obliges one to calculate, by means of the integral, the heat dissipated.

Equilibrium temperature

It should be noted [1] that eqn. (3) is only useful if the heat capacity displays a small change between the start and the end of the reaction. This only occurs if the system is in thermal equilibrium during the initial and final periods. For this, at the beginning of the program the temperature T_∞ , at which the system is in perfect thermal equilibrium, is required. The new value of T_∞ is also required when a new run of data must be inserted (step 4).

Modification of data for time–voltage pairs

This subroutine allows the following operations: (1) change of time–voltage pair; (2) addition of time–voltage pair; (3) deletion of time–voltage pair.

This subroutine is very useful for steps (1)–(3) of the menu, when data input are digitized.

Bubble sort

As regards the above-cited operations, this subroutine is able to re-organize the time–voltage pairs order as a function of increasing time.

Change of reaction data

With this subroutine it is possible to execute again the calculation of the heats of reaction, changing the parameters K/C , instantaneous power and mole number of the reagents.

REMARKS ON THE PROGRAM

Reaction enthalpy program test

For the ideal adiabatic calorimeter the corrective term $K/C \int_{t_0}^t (T - T_\infty) dt$ approaches zero. This is partially true for the Tronac calorimeter. Anyway its corrective term values are usually smaller than those of the LKB calorimeter. The behaviour is observed for the standard thermochemical reaction values between solid tris(hydroxymethyl)amino methane and 0.1 M HCl at 25 °C in water. The value obtained (see Table 1) is 29.712 kJ and was compared with that obtained in the literature [11] of 29.744 kJ. The average percentage deviation of the former relative to the latter is 0.10%.

K/C value choice

As previously shown [2], for the constant-environment calorimeter, the choice of K/C parameter is fundamental for a correct calculation of the

TABLE 1

Partial molar enthalpy of reaction ΔH between solid tris(hydroxymethyl)amino methane and 0.1 M HCl(aq) in water at 25 °C

tris (g)	$-\Delta H$ (kJ mol ⁻¹)
0.098	29.400
0.132	29.538
0.144	30.028
0.787	29.922
0.981	29.700
0.775	30.100
0.123	29.300

temperature-corrective term. Indeed, it is possible to calculate two values for the K/C coefficient: one for the calibration cooling curve and another for the exothermic reaction cooling curve.

We have observed [2] that for a “fast” reaction it is convenient to use the K/C value of the cooling calibration curve, while for a “slow” exothermic reaction that of the reaction cooling curve is more useful.

We have found that the two values of K/C must differ by no more than 4–5%: this usually occurs when the system reaches perfect thermal equilibrium.

For the quasi-adiabatic system the differences in K/C values for “fast” and “slow” reactions are significantly smaller than those of the constant-environment calorimeter. This is due to the smaller leakage constant value of the adiabatic calorimeter with respect to that of the constant-environment calorimeter. The consequently smaller (70%) K/C term renders the quasi-adiabatic calorimeter less liable to thermal equilibrium variation than the constant-environment one.

End-point of the reaction

The end-point of the reaction is characterized by the change in sign of the partial molar enthalpy of the reaction. In the constant-environment calorimeter this point is easily read for “fast” reactions, while it can hardly be read for a very “slow” reaction. Indeed, for the farthest pairs, the mathematical errors introduced by the temperature correction term are larger than the corresponding calculated heats. Furthermore, in the long run the thermal equilibrium condition would be changed so that the reaction $\Delta T_{\text{obs}} = -K/C \int_{T_{\infty}}^T (T - T_{\infty}) dt$ would no longer be effective. For the adiabatic calorimeter, this does not occur because the larger temperature variation due to the smaller K/C value supplies a greater “sensitiveness” for the enthalpy values reading.

Appendix

```

1 REM
2 REM a basic computer program of calculation for the determination
3 REM of solution and reaction heats in a quasi-adiabatic calorimeter
10 FALSE% = 0 : TRUE% = NOT FALSE : MAXNUM# = 1.701411834604692D+38
20 M# = .000003169#:Q# = 25# : REM Parameters used on subroutine 10000 for volta
    ge to temperature conversion.You must determine
    the parameters as specified in the article
30 FOR I% = 1 TO 10 : KEY I%,"" : NEXT
40 KEY OFF : VIEW PRINT : CLS : LPRINT
50 ON KEY (1) GOSUB 22000 : ON KEY (2) GOSUB 23000
60 N%=1000
70 DIM VOLT#(N%) , SECOND#(N%) , TEMP#(N%)
80 PRINT "EQUILIBRIUM TEMPERATURE VOLTAGE (hundredths of mV)";INPUT VC#
90 MILLI.VOLT# = VC# : GOSUB 10000 : EQUIL.TEMP# = TEMPERATURE#
100 REM Temperature at thermal equilibrium before calibration
110 CLS : LOCATE 2
120 PRINT TAB(15) ; "CALCULATIONS FOR TRONAC CALORIMETER"
130 LOCATE 5
140 PRINT : PRINT TAB(20) ; "1 - CALIBRATION CURVE"
150 PRINT : PRINT TAB(20) ; "2 - COOLING CURVE"
160 PRINT : PRINT TAB(20) ; "3 - REACTION CURVE"
170 PRINT : PRINT TAB(20) ; "4 - EQUILIBRIUM TEMPERATURE"
180 PRINT : PRINT TAB(20) ; "5 - END" : PRINT
190 LOCATE 19,20 : INPUT"OPTION (1-5)";OP%
200 IF OP% < 1 OR OP% > 5 THEN PRINT CHR$(7); : GOTO 190
210 IF OP% = 5 THEN CLS : SYSTEM
220 IF OP% = 4 THEN CLS : GOTO 80
230 ON OP% GOSUB 1000 , 2000 ,3000
240 GOTO 110
1000 REM *****
1010 REM          CALIBRATION
1020 REM *****
1030 LINEO$ ="          DATA FOR CALIBRATION
    " : I% = 0 : GOSUB 20000
1040 FOR K% = 1 TO I%
1050 MILLI.VOLT# = VOLT#(K%) : GOSUB 10000 : TEMP#(K%) = TEMPERATURE# - EQUIL.TE
MP#
1060 NEXT
1070 REM Lines 1080 to 1210 : calculation and printing of dT/dt
1080 TIME.SUM# = 0# : SQUARE.TIME.SUM# = 0# : TIME.X.TEMP.SUM# = 0# : TEMP.SUM#
= 0#
1090 LPRINT STRING$(33,"*") ; " CALIBRATION " ; STRING$(34,"*") : LPRINT
1100 LPRINT "SECONDS" ; TAB(20) ; "VOLTS" ; TAB(40) ; "TEMPERATURE"
1110 FOR K% = 1 TO I%
1120 LPRINT SECOND#(K%) ; TAB(20) VOLT#(K%) ; TAB(40) ;
1130 LPRINT USING "##.#####"; TEMP#(K%)
1140 TIME.SUM# = TIME.SUM# + SECOND#(K%)
1150 SQUARE.TIME.SUM# = SQUARE.TIME.SUM# + SECOND#(K%)^2#
1160 TIME.X.TEMP.SUM# = TIME.X.TEMP.SUM# + SECOND#(K%)*TEMP#(K%)
1170 TEMP.SUM# = TEMP.SUM# + TEMP#(K%)
1180 NEXT K%
1190 TEMP.VAR.SPEED# = (I% * TIME.X.TEMP.SUM# - TIME.SUM# * TEMP.SUM#)/(I% * SQ
UARE.TIME.SUM# - TIME.SUM#^2#)
1200 LPRINT
1210 LPRINT "dT/dt = ";TEMP.VAR.SPEED#
1220 AVG.TEMP# = TEMP.SUM# / I%
1230 LPRINT
1240 LPRINT "AVERAGE TEMPERATURE OF CALIBRATION =" ; AVG.TEMP#
1250 LPRINT STRING$(37,"*") ; " END " ; STRING$(38,"*") :LPRINT : LPRINT
1260 GOSUB 30000
1270 IF AGAIN% THEN GOTO 1040 ELSE RETURN
2000 REM *****
2010 REM          COOLING

```

```

2020 REM *****
2030 REM cooling for esothermic or heating for endothermic reactions
2040 LINEO$ ="          DATA FOR COOLING
      " : I% = 0 : GOSUB 20000
2050 LPRINT STRING$(35,"*") ; " COOLING " ; STRING$(36,"*") : LPRINT
2060 LPRINT "TEMPERATURE AT EQUILIBRIUM = " ; EQUIL.TEMP# : LPRINT
2070 LPRINT "PAIR"; TAB(10);"SECONDS";TAB(30);"VOLTS";TAB(50)"DELTA T":LPRINT
2080 FOR K% = 1 TO I%
2090 MILLI.VOLT# = VOLT#(K%) : GOSUB 10000 : TEMP#(K%) = TEMPERATURE# - EQUIL.TE
MP#
2100 LPRINT K% ; TAB(10) ; SECOND#(K%) ; TAB(30) ; VOLT#(K%) ; TAB(50) ;
2110 LPRINT USING "###.#####"; TEMP#(K%)
2120 NEXT K%
2130 LPRINT : LPRINT"PAIR" ; TAB(20); " K/C" : LPRINT
2140 REM Lines 2150 to 2230 : Calculation and printing of k/c and average k/c
      See formula 17 on reference 1 in the article

2150 SUM# = 0#; COUNT% = 0
2160 FOR K% = 1 TO I% - 1
2170 FOR L% = K% + 1 TO I%
2180 K.ON.C# = LOG(TEMP#(L%) / TEMP#(K%)) / (SECOND#(K%) - SECOND#(L%))
2190 LPRINT K% ; " -"; L% ; TAB(20) ; : LPRINT USING "###.#####"; K.ON.C#
2200 SUM# = SUM# + K.ON.C# : COUNT% = COUNT% + 1
2210 NEXT L%
2220 NEXT K%
2230 LPRINT : LPRINT "AVERAGE K/C =" ; SUM# / COUNT% : LPRINT
2240 LPRINT STRING$(37,"*") ; " END " ; STRING$(38,"*") :LPRINT : LPRINT
2250 GOSUB 30000
2260 IF AGAIN% THEN GOTO 2060 ELSE RETURN
3000 REM *****
3010 REM          REACTION
3020 REM *****
3030 LINEO$ ="          DATA FOR REACTION
      " : I% = 0 : GOSUB 20000
3040 INPUT "K/C =" ; CHOSEN.K.ON.C#
3050 INPUT "NUMBER OF MOLE =" ; MOLE#
3060 INPUT "CALIBRATION CURRENT (mA) =" ; CAL.CURR#
3070 INPUT "CALIBRATION VOLTAGE (V) =" ; CAL.VOLT#
3080 CAL.POWER# = CAL.VOLT# * CAL.CURR# / 1000# : REM See formula 18 on
      reference 1
3090 THERM.CAPACITY# = CAL.POWER# / (TEMP.VAR.SPEED# + CHOSEN.K.ON.C# * AVG.TEMP
#)
3100 LPRINT STRING$(35,"*") ; " REACTION " ; STRING$(35,"*") : LPRINT
3110 LPRINT "TEMPERATURE AT EQUILIBRIUM =" ; EQUIL.TEMP#
3120 LPRINT "CALIBR. VOLTAGE =" ; CAL.VOLT#,"CALIBR. CURRENT =" ; CAL.CURR#
3130 LPRINT "CHOSEN K/C =" ; CHOSEN.K.ON.C# , "MOLE =" ; MOLE#
3140 LPRINT "CALCULATED THERMAL CAPACITY =" ; THERM.CAPACITY#
3150 LPRINT : LPRINT
3160 LPRINT "PAIR"; TAB(10); "SECONDS" ; TAB(30) ; "VOLTS" ; TAB(50) "DELTA T " :
LPRINT
3170 FOR K% = 1 TO I%
3180 MILLI.VOLT# = VOLT#(K%) : GOSUB 10000 : TEMP#(K%) = TEMPERATURE# - EQUIL.TE
MP#
3190 LPRINT K% ; TAB(10) ; SECOND#(K%) ; TAB(30) ; VOLT#(K%) ; TAB(50) ;
3200 LPRINT USING "###.#####"; TEMP#(K%)
3210 NEXT K%
3220 REM Lines 3230 to 3400 : calculation and printing of delta_H , increase
      of delta_H and increase of corrective delta_H

3230 INTEGRAL# = 0#
3240 PREVIOUS.DELTA.H# = 0#
3250 LPRINT : LPRINT "FROM POINT" ; TAB(43) "INCREASE OF" ; TAB(65) "INCREASE OF"
3260 LPRINT" 1 TO ..."; TAB(20) ; " DELTA H" ; TAB(45) ; "DELTA H" ; TAB(62) "C
ORRECTIVE DELTA H" : LPRINT
3270 FOR K% = 2 TO I%

```



```

3280 INCRINT# = (SECOND#(K%) - SECOND#(K%-1)) * ((TEMP#(K%) + TEMP#(K%-1)) / 2#)
3290 INTEGRAL# = INTEGRAL# + INCRINT#
3300 DELTA.T.OBS# = TEMP#(K%) - TEMP#(1)
3310 DELTA.T.CORR# = CHOSEN.K.ON.C# * INTEGRAL# : REM See formula 8 on
                                         reference 1
3320 DELTA.T.ADB# = (DELTA.T.OBS# + DELTA.T.CORR#) : REM See formula 7 on
                                         reference 1
3330 DELTA.H# = DELTA.T.ADB# * (THERM.CAPACITY# / (MOLI# * 1000#))
3340 INCR.DELTA.H# = DELTA.H# - PREVIOUS.DELTA.H#
3350 PREVIOUS.DELTA.H# = DELTA.H#
3360 INCR.CORR.DELTA.H# = CHOSEN.K.ON.C# * INCRINT# * THERM.CAPACITY# / (MOLI# *
1000#)
3370 LPRINT TAB(4) K% ; TAB(14) ; : LPRINT USING "#####.#####"; DELTA.H#;
3380 LPRINT TAB(40) ; : LPRINT USING "#####.#####"; INCR.DELTA.H#;
3390 LPRINT TAB(63) ; : LPRINT USING "#####.#####"; INCR.CORR.DELTA.H#
3400 NEXT K%
3410 LPRINT STRING$(37,"*"); " END "; STRING$(38,"*"):LPRINT : LPRINT
3420 PRINT : INPUT "WOULD YOU EXECUTE AGAIN THE CALCULIION CHANGING THE PARAMETE
RS (Y/N) ";YES$
3430 IF YES$ = "Y" OR YES$ = "y" GOTO 3040
3440 IF YES$ <> "N" AND YES$ <> "n" GOTO 3420
3450 GOSUB 30000
3460 IF AGAIN% THEN GOTO 3040 ELSE RETURN
10000 REM *****
10010 REM VOLTAGE -> TEMPERATURE CONVERSION ROUTINE
10020 REM *****
10030 MILLI.VOLT# = MILLI.VOLT# * 100# : REM NOTICE
                                         This routine is the only one that
10040 TEMPERATURE# = M# * MILLI.VOLT# + Q#:REM must be changed if the voltage vs
                                         temperature relationship of your
                                         calorimeter is not linear
10050 RETURN
20000 REM *****
20010 REM VOLTAGE-TIME PAIRS INTRODUCTION ROUTINE
20020 REM *****
20030 KEY (1) ON : KEY (2) ON :LOGIC% = 0
20040 CLS : LINE1$ = " F1 + CR : RETURN TO PREVIOUS QUESTION" : LINE2$ = " F2
+ CR : END OF INTRODUCTION" : GOSUB 21000
20050 I% = I% + 1
20060 PRINT "TIME NUMBER "; I% ; " ( seconds ) " ; : INPUT SECOND#
(I%) : I% = I%
20070 IF LOGIC% = 1 THEN LOGIC% = 0 : I% = I% - 1 : IF I% = 0 THEN I% = 1 : PRINT
CHR$(7) : GOTO 20060 ELSE PRINT : GOTO 20090
20080 IF LOGIC% = 2 THEN LOGIC% = 0 : I% = I% - 1 : GOTO 20130
20090 PRINT "VOLTAGE NUMBER "; I% ; " ( hundredths of mV ) " ; : INPUT VOLT#(I
%) : I% = I%
20100 IF LOGIC% = 1 THEN LOGIC% = 0 : PRINT : GOTO 20060
20110 IF LOGIC% = 2 THEN LOGIC% = 0 : I% = I% - 1 : GOTO 20130
20120 PRINT : IF I% < N% GOTO 20050
20130 VIEW PRINT : CLS : KEY (1) OFF : KEY (2) OFF : GOSUB 40000 : RETURN
21000 REM *****
21010 REM WINDOWS
21020 REM *****
21030 VIEW PRINT 1 TO 1
21040 PRINT LINE0$;
21050 VIEW PRINT 24 TO 25
21060 PRINT LINE1$
21070 PRINT LINE2$;
21080 VIEW PRINT 2 TO 23
21090 RETURN
22000 REM *****
22010 REM F1 DETECTION
22020 REM *****

```

```

22030 LOGIC% = 1
22040 RETURN
23000 REM *****
23010 REM           F2 DETECTION
23020 REM *****
23030 LOGIC% = 2
23040 RETURN
30000 REM *****
30010 REM           NEW DATA ROUTINE
30020 REM *****
30030 CLS : KEY (1) ON
30040 INPUT "DO YOU WANT TO MODIFY ANY DATA ? (Y/N)";YES$
30050 IF YES$ = "N" OR YES$ = "n" THEN AGAIN% = FALSE% : RETURN
30060 IF YES$ <> "Y" AND YES$ <> "y" THEN GOTO 30040
30070 AGAIN% = TRUE%
30080 CLS : PRINT TAB(20) "MODIFICATION OF DATA" : LOCATE 5
30090 PRINT : PRINT TAB(20) " 1 - CHANGE OF A PAIR"
30100 PRINT : PRINT TAB(20) " 2 - ADDITION OF A PAIR"
30110 PRINT : PRINT TAB(20) " 3 - DELETION OF A PAIR"
30120 PRINT : PRINT TAB(20) " 4 - END OF MODIFICATION"
30130 LOCATE 19,20 : INPUT "OPTION (1 - 4)";OP%
30140 IF OP% < 1 OR OP% > 4 THEN GOTO 30130
30150 IF OP% = 4 THEN CLS : KEY (1) OFF : RETURN
30160 ON OP% GOSUB 31000 , 32000 , 33000
30170 GOSUB 40000
30180 VIEW PRINT : CLS : PRINT : GOTO 30080
31000 REM *****
31010 REM           PAIRS CHANGING
31020 REM *****
31030 CLS : KEY (1) ON
31040 LINE0$ = "                               CHANGE OF PAIRS
      "
31050 LINE1$ = " F1 + CR : END OF CHANGING" : LINE2$ = "" : GOSUB 21000
31060 LOCATE 5,1 : PRINT SPC(60) : LOCATE 5,1 : PRINT "PAIR NUMBER ( 1 -"; I% ;
      " ) " ; : INPUT PAIR : REM LEAVE IT
31070 IF LOGIC% = 1 THEN LOGIC% = 0 : RETURN
31080 IF PAIR < 1 OR PAIR > I% GOTO 31060
31090 LOCATE 8 : PRINT TAB(20) "TIME" ; TAB(40) "VOLTAGE"
31100 PRINT : PRINT
31110 PRINT "OLD PAIR";TAB(20); SECOND#(PAIR) ; TAB(40) ; VOLT#(PAIR)
31120 PRINT : PRINT
31130 PRINT "NEW PAIR"; TAB(20) : INPUT ; "" , COM1# : REM LEAVE IT
31140 IF LOGIC% = 1 THEN LOGIC% = 0 : RETURN ELSE SECOND#(PAIR) = COM1#
31150 PRINT TAB(40) : INPUT "" , COM2# : REM LEAVE IT
31160 IF LOGIC% = 1 THEN LOGIC% = 0 : RETURN ELSE VOLT#(PAIR) = COM2#
31170 GOTO 31030
32000 REM *****
32010 REM           ADDITION ROUTINE
32020 REM *****
32030 LINE0$ = "                               ADDITION OF PAIRS
      " : GOSUB 20000
32040 KEY (1) ON
32050 RETURN
33000 REM *****
33010 REM           DELETION ROUTINE
33020 REM *****
33030 CLS : LINE0$ = "                               DELETION OF PAIRS
      "
33040 LINE1$ = " F1 + CR : END OF DELETION " : LINE2$ = "" : GOSUB 21000
33050 DELETED = 0
33060 LOCATE 5,20 : PRINT SPC(60) : LOCATE 5,20 : PRINT "PAIR NUMBER ( 1 -"; I% ;
      " ) " ; : INPUT PAIR : REM LEAVE IT
33070 IF LOGIC% = 1 THEN LOGIC% = 0 : GOSUB 40000 : I% = I% - DELETED : RETURN

```

```

33080 IF PAIR < 1 OR PAIR > I% GOTO 33060
33090 IF SECOND#(PAIR) <> MAXNUM# THEN GOTO 33120
33100 LOCATE 12,20 : PRINT "PAIR ALREADY DELETED" : PRINT : PRINT : PRINT "PRESS
      ANY KEY TO CONTINUE " ;
33110 GHOST$ = INPUT$(1) : CLS : GOTO 33060
33120 LOCATE 8 : PRINT TAB(20) "TIME" ; TAB(40) "VOLTAGE"
33130 PRINT : PRINT
33140 PRINT TAB(20) SECOND#(PAIR) ; TAB(40) VOLT#(PAIR)
33150 LOCATE 15 : INPUT "DO YOU WANT TO DELETE (Y - N)";YES$
33160 IF YES$ = "N" OR YES$ = "n" THEN CLS : GOTO 33060
33170 IF YES$ <> "Y" AND YES$ <> "y" THEN GOTO 33150
33180 SECOND#(PAIR) = MAXNUM# : DELETED = DELETED + 1 : CLS : GOTO 33060
40000 REM *****
40010 REM          BUBBLE SORT
40020 REM *****
40030 UP% = 1
40040 SORTED% = FALSE%
40050 WHILE NOT SORTED%
40060 SORTED% = TRUE%
40070 FOR K% = I% TO UP% STEP -1
40080 IF SECOND#(K%)<SECOND#(K% - 1) THEN SWAP SECOND#(K%),SECOND#(K% - 1):SWAP
      VOLT#(K%),VOLT#(K% - 1):SORTED%=FALSE%
40090 NEXT K%
40100 UP% = UP% + 1
40110 WEND
40120 RETURN

```

REFERENCES

- 1 F. Rodante and A. Onofri, *Thermochim. Acta*, 94 (1985) 239.
- 2 F. Rodante and R. Rosati, *Thermochim. Acta*, 117 (1987) 167.
- 3 J.J. Christensen, J. Ruckman, D.J. Eatough and R.M. Izzat, *Thermochim. Acta*, 3 (1972) 203.
- 4 D.J. Eatough, R.M. Izatt and J.J. Christensen, *Thermochim. Acta*, 3 (1972) 219.
- 5 D.J. Eatough, R.M. Izzat and J.J. Christensen, *Thermochim. Acta*, 3 (1972) 233.
- 6 D.J. Eatough, J.J. Christensen, R.M. Izatt, *Experiments in Thermometric Titrimetry and Titration Calorimetry*, Brigham Young University Press, Provo, Utah, 1974.
- 7 Rahmalan Bin Ahmed, J.O. Hill and R.J. Magee, *Thermochim. Acta*, 71 (1983) 25.
- 8 V. Cerda, J.M. Estela and R. Jara, *Thermochim. Acta*, 87 (1985) 13.
- 9 F. Gaizer and A. Puskas, *Talanta*, 28 (1981) 925.
- 10 F. Gaizer and A. Puskas, *Talanta*, 28 (1981) 565.
- 11 J.O. Hill, G. Ojelund and I. Wadso, *J. Chem. Thermodyn.*, 1 (1969) 111.